LABORATORIO DI MATEMATICA
I NUMERI COMPLESSI CON EXCEL

ESERCITAZIONE GUIDATA

Costruiamo un foglio elettronico che, letti la parte reale a e il coefficiente b della parte immaginaria, calcoli le n radici n-esime del numero complesso $a + bi$.
Troviamo con il foglio le tre radici cubiche di i ($a = 0$ e $b = 1$).

L'analisi del problema

Per ricavare le radici n-esime di un numero complesso $a + bi$ ricorriamo alla formula di De Moivre $c_k = \sqrt[n]{r}\left[\cos\left(\frac{\alpha}{n} + \frac{2k\pi}{n}\right) + i\cos\left(\frac{\alpha}{n} + \frac{2k\pi}{n}\right)\right]$ con $k = 0, 1, \ldots, n - 1$, calcolandone i vari termini.

In particolare, otteniamo l'argomento α operando i seguenti controlli su a e su b:

se $a = 0$ e $b < 0$, $\alpha = \frac{3}{2}\pi$; se $a = 0$ e $b \geq 0$, $\alpha = \frac{1}{2}\pi$;

se $a > 0$ e $b \geq 0$, $\alpha = \operatorname{arctg}\frac{b}{a}$; se $a > 0$ e $b < 0$, $\alpha = \operatorname{arctg}\frac{b}{a} + 2\pi$;

se ($a < 0$ e $b \geq 0$) o ($a < 0$ e $b < 0$), $\alpha = \operatorname{arctg}\frac{b}{a} + \pi$.

La costruzione del foglio

- Scriviamo delle indicazioni e mettiamo dei bordi alle celle B4, D4 e I4 per segnalare dove immettere i valori dei coefficienti a e b e l'indice n della radice e inseriamo delle didascalie per leggere i risultati (figura 1).
- In B6 digitiamo =RADQ(B4^2 + D4^2) per il calcolo di $r = \sqrt{a^2 + b^2}$.
- In E6 inseriamo la formula con i condizionali per determinare α, =SE(B4 = 0; SE(D4 < 0; 3*PI.GRECO()/2; PI.GRECO()/2); SE(B4 > 0; SE(D4 > 0; ARCTAN(D4/B4); ARCTAN(D4/B4) + 2*PI.GRECO()); ARCTAN(D4/B4) + PI.GRECO())).
- In F6, trasformiamo α in gradi con =E6*180/PI.GRECO().
- Calcoliamo la radice aritmetica n-esima di r con =B6^(1/I4) in B7.
- In E7 scriviamo =E6/I4 per calcolare $\theta = \frac{\alpha}{n}$.
- In F7, trasformiamo θ in gradi con =E7*180/PI.GRECO().
- In I7, digitiamo =2*PI.GRECO()/I4 per ricavare $\Delta\alpha = \frac{2\pi}{n}$.
- In J7, trasformiamo $\Delta\alpha$ in gradi scrivendo =I7*180/PI.GRECO().
- Per sviluppare le radici dalla formula di De Moivre, mettiamo un contatore nella colonna A che parta da 0 e si fermi a $n - 1$; in A10 digitiamo pertanto 0 e in A11 =SE(O(A10 = I4 − 1; A10 = "/"); "/"; A10 + 1) e la copiamo sino alla A14 (rendiamo atta la tabella a contenere sino alle radici quinte).
- Nelle seguenti celle scriviamo le formule per rappresentare le n radici sotto forma trigonometrica e sotto forma algebrica. In ognuna inseriamo un controllo riferito alla colonna A, che, se l'indice $n - 1$ è superato, lasci la cella vuota.

```
B10  =SE(A10 = "/"; ""; $B$7)
D10  =SE(A10 = "/"; ""; $F$7 + A10*$J$7)
F10  =SE(A10 = "/"; ""; $F$7 + A10*$J$7)
H10  =SE(A10 = "/"; ""; $B$7*COS($E$7 + A10*$I$7))
```

J10 =SE(A10 = "/"; ""; B7*SEN(E7 + A10*I7))
C10 =SE(A10 = "/"; ""; "* (COS(")
E10 =SE(A10 = "/"; ""; ") + i* SEN(")
G10 =SE(A10 = "/"; ""; "))")
I10 =SE(A10 = "/"; ""; " + i*")

- Copiamo quindi la zona B10:J10 sino alla riga 14.
- Mettiamo dei bordi alle tabelle delle radici.

L'applicazione del foglio

- Per trovare le radici cubiche di *i*, in B4 digitiamo 0, in D4 1 e in I4 3.

	A	B	C	D	E	F	G	H	I	J
1	Le radici di un numero complesso									
2										
3	Inserisci i coefficienti del numero complesso a + b i e l'indice della radice									
4	a =	0,0000	b =	1,0000				n =	3	
5										
6	r =	1,0000		α =	1,5708	90				
7	r^(1/n) =	1,0000		α/n =	0,5236	30		Δ di α =	2,0944	120
8										
9	k	Le radici sotto forma trigonometrica						Le radici sotto forma algebrica		
10	0	1,0000	* (COS(	30	)+ i*SEN	30	))	0,8660	+ i *	0,5000
11	1	1,0000	* (COS(	150	)+ i*SEN	150	))	-0,8660	+ i *	0,5000
12	2	1,0000	* (COS(	270	)+ i*SEN	270	))	0,0000	+ i *	-1,0000
13	/									
14	/									

▲ **Figura 1** Il foglio con le radici cubiche di *i*.

Esercitazioni

Nei seguenti esercizi, costruisci un foglio che, letti la parte reale e il coefficiente della parte immaginaria dei dati indicati, svolga quanto è richiesto.

1 Dati due numeri complessi, determini i tre coefficienti di un'equazione di secondo grado che abbia come radici i due numeri dati.
Nel foglio inserisci anche la verifica.

2 Forniti sei numeri complessi, coefficienti di un sistema lineare con due equazioni e due incognite, determini il tipo di sistema e, se è determinato, ne trovi la soluzione.
Nel foglio inserisci anche la verifica.

3 Letti un numero complesso c, un numero naturale $n < 5$ e gli $n + 1$ numeri complessi a_i coefficienti del polinomio P_n di grado n, calcoli il valore del polinomio quando l'indeterminata assume il valore c. Amplia il foglio per calcolare i valori del polinomio corrispondenti a valori di c variabili in una zona del piano di Gauss.

4 Date n coppie di numeri complessi, mostri che soddisfano la disuguaglianza triangolare.

5 Forniti due numeri complessi, calcoli la loro somma e rappresenti graficamente nel piano di Gauss i vettori corrispondenti agli addendi e quello corrispondente alla somma.